

NAG Fortran Library Routine Document

X04ADF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

X04ADF closes a file associated with a given Fortran unit number.

2 Specification

```
SUBROUTINE X04ADF(IUNIT, IFAIL)
INTEGER          IUNIT, IFAIL
```

3 Description

X04ADF is especially useful if the calling language is not Fortran. It closes a file associated with a given Fortran unit number.

4 References

None.

5 Parameters

1: IUNIT – INTEGER *Input*
On entry: the Fortran unit number which identifies the file to be closed.

2: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, –1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or –1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1
Failure to close the file.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example program simply illustrates how to close a file once it has been opened for writing followed by how to close a file once it has been opened for reading.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      X04ADF Example Program Text.
*      Mark 19 Release. NAG Copyright 1999.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER       (NOUT=6)
      INTEGER          ICHAN
      PARAMETER       (ICHAN=4)
      CHARACTER*11    FNAME
      PARAMETER       (FNAME='success.res')
*      .. Local Scalars ..
      INTEGER          IFAIL
      LOGICAL          EX, OP
*      .. External Subroutines ..
      EXTERNAL        X04ACF, X04ADF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'X04ADF Example Program Results'
*
*      Test successful open and close for write
*
      IFAIL = 1
*
      CALL X04ACF(ICHAN,FNAME,1,IFAIL)
*
      INQUIRE (ICHAN,EXIST=EX,OPENED=OP)
      IF ((IFAIL.EQ.0) .AND. EX .AND. OP) THEN
          WRITE (NOUT,99998)
          WRITE (ICHAN,99998)
      ELSE
          WRITE (NOUT,99999) FNAME, ICHAN
      END IF
*
      IFAIL = 1
*
      CALL X04ADF(ICHAN,IFAIL)
*
      INQUIRE (ICHAN,OPENED=OP)
      IF ((IFAIL.EQ.0) .AND. (.NOT. OP)) THEN
          WRITE (NOUT,99997)
      ELSE
          WRITE (NOUT,99996) FNAME, ICHAN
      END IF
*
*      Test successful open and close for read
*
      IFAIL = 1
*
      CALL X04ACF(ICHAN,FNAME,0,IFAIL)
*
      INQUIRE (ICHAN,EXIST=EX,OPENED=OP)
      IF ((IFAIL.EQ.0) .AND. EX .AND. OP) THEN
          WRITE (NOUT,99994)
      ELSE
          WRITE (NOUT,99995) FNAME, ICHAN
      END IF
```

```
*
      IFAIL = 1
*
      CALL X04ADF(ICHAN,IFAIL)
*
      INQUIRE (ICHAN,OPENED=OP)
      IF ((IFAIL.EQ.0) .AND. ( .NOT. OP)) THEN
        WRITE (NOUT,99997)
      ELSE
        WRITE (NOUT,99996) FNAME, ICHAN
      END IF
*
      STOP
*
99999 FORMAT (' ** FAILS to open "',A,'" and connect file to channel ',
+           I2)
99998 FORMAT (' OK file successfully opened for writing')
99997 FORMAT (' OK file successfully closed')
99996 FORMAT (' ** FAILS to close "',A,'" on channel ',I2)
99995 FORMAT (' ** FAILS to open "',A,'" for reading on channel ',I2)
99994 FORMAT (' OK file successfully opened for reading')
      END
```

9.2 Program Data

None.

9.3 Program Results

```
X04ADF Example Program Results
OK file successfully opened for writing
OK file successfully closed
OK file successfully opened for reading
OK file successfully closed
```
